

## ANALYTIC GEOMETRY OF THE PLANE AND MATHEMATICA

Maja Bator, Zvonko Čerin, Milena Čulav (Zagreb, Croatia)

**Abstract.** We describe the use of the program Mathematica in the analytic geometry of the plane in the rectangular coordinates. As an illustration of possible applications of this method we present the solutions of fifteen problems from the problem book for the first class of gymnasiums in Croatia.

**AMS Classification Number:** 00A35

### 1. Introduction

In this article we describe the computer approach to the analytic geometry of the plane. In order to do this we shall use the symbolic computation program **Mathematica**. Of course, the same could be done in the rival program **Maple V**. These are the most widely known and the most popular extensive systems or CAS that “know mathematics”. Each of them has its own programming language. Our task is reduced to describing basic functions that are needed for solving geometry problems with the analytic method.

This is the translation to English of the article [2] that is in Croatian. In the references [1], [3] and [4] that are also in Croatian the same task was done in the program Maple V. The whole project is the result of the second author’s course “Geometry and computers” at the Mathematics Department of the University of Zagreb (in Croatia) in which the first and the third authors (the undergraduate mathematics teachers students) have been enrolled in the academic year 2002/2003.

This elective course is offered to all fourth year mathematics major students. The number of students is growing so that for the academic year 2004/2005 there will be ten participants. The aim of the course is to teach how to use computers in mathematics working on projects under the guidance of the professor. We meet four hours each week in the computer laboratory. The first few weeks the professor is presenting the basics of text processing (LaTeX) in the program WinEdt and the commands in Mathematica and Maple. For figures in geometry we use the Geometer’s Sketchpad. None of these programs is really explained in all details because we believe that they could be helpful even if we have rather limited knowledge of them just as we drive cars without being mechanics. The students

pick up on their own more advanced features of these programs later on while working on the project.

What the project can be will become clear in the rest of this article because this is an example of the final outcome. In short, here the project was to program functions in Mathematica which cover analytic geometry in the plane and use them to solve with computers several problems from the secondary school level as we wanted to publish this in the Croatian mathematics and physics journal for high schools “Matematičko–fizički list”. Some other projects were geometric inequalities, properties of regular polygons, and identities for Fibonacci and Lucas numbers.

All this effort is in the direction to help teachers in Croatia to accept computers as an important tool in teaching mathematics. The Croatian Mathematical Society has started an experimental program for two groups of the first and the third year pupils of gymnasiums in Zagreb that could be described as mathematics with computers. Both high school and university professors are involved in this effort but a lot of work still remains to effectively introduce computers into all levels of schools. Ours is only a small contribution on this way.

## 2. Basic function of analytic geometry

The key idea of the analytic geometry is to associate algebraic entities with geometric objects and then investigate them using algebraic methods.

The input of points on the plane in Maple V is quite simple because they are just ordered pairs of real numbers (their rectangular coordinates). For example, the input

```
tA:={2, 3}; tB:={5, 7}; tC:={-2, 0}; tT:={x, y};
```

defines four points on the plane  $A(2, 3)$ ,  $B(5, 7)$ ,  $C(-2, 0)$ ,  $T(x, y)$ .

The function `FS` is a shortcut for the simultaneous use of commands `Factor` and `FullSimplify` while `distance` measures the distance.

```
FS[m_]:=Factor[FullSimplify[m]]
```

```
distance[{a_,u_}, {b_,v_}]:=Sqrt[FS[(b-a)^2+(v-u)^2]]
```

The name of this function is *distance*. It asks for two ordered pairs of real numbers. The first pair has the components  $a$  and  $u$  while the components of the second pair are  $b$  and  $v$ . The machine first computes  $(b - a)^2 + (v - u)^2$  and then tries as much as possible to simplify and factor this sum of squares (the command `FS`). In the end it finds the square root of everything (the command `Sqrt`).

Many times it is important to determine the *midpoint* of the segment whose endpoints are given or the point which divides this segment either in *ratio*  $k$  (real number different from  $-1$ ) or in the ratio  $\frac{m}{n}$  (of real numbers whose sum is not zero).

```
midpoint[{a_,u_}, {b_,v_}]:=FS[{(a+b)/2,(u+v)/2}]
```

```
ratio[{a_,u_},{b_,v_},k_]:=FS[{(a+k*b)/(1+k),(u+k*v)/(1+k)}]
rationm[{a_,u_},{b_,v_},m_,n_]:=FS[{(a*n+b*m)/(m+n),(u*n+v*m)/(m+n)}]
```

The lines in the program Mathematica are represented as ordered triples  $[a, b, c]$  of coefficients of their linear equations. For example, the input

```
pX:={1,0,0}; pY:={0,1,0}; pD:={1,-1,0}; pG:={-1,2,2}
```

defines four lines in the plane. They are the  $y$ -axis, the  $x$ -axis, the bisector of the first and the third quadrant and the line  $-x + 2y + 2 = 0$ .

The line is given either by one of its points and the tangent  $k$  of the angle which it makes with the positive direction of the  $x$ -axis (better known as its *slope*) or by two different points.

```
line1[k_,{b1_,b2_}]:=FS[{k,-1,b2-b1*k}]
line2[{x1_,y1_},{x2_,y2_}]:=FS[{y2-y1,x1-x2,x1*y2-x2*y1}]
```

Sometimes it is useful to have the following functions which test if a point lies on a line and if three points are collinear. The letter Q in their names suggests the word“question”. A point is on a line or points are collinear if and only if the function evaluates to zero.

```
onlineQ[{a_,b_},{x_,y_,z_}]:=FS[a*x+y*b+z]
collinearQ[{x1_,y1_},{x2_,y2_},{x3_,y3_}]:=
  FS[y2*x3-y1*x3+x1*y3-x2*y3+x1*y2-x2*y1]
```

The intersection of lines or the information that they are parallel (when we get the error message of division with zero) gives our next function.

```
inter[{a_,b_,c_},{i_,j_,k_}]:=
  FS[(-j*c+k*b)/(-i*b+a*j),(i*c-a*k)/(-i*b+a*j)]
```

Functions for the parallel and the perpendicular through a point to a line and tests if lines are parallel or perpendicular are next.

```
parallel[{a_,b_},{x_,y_,z_}]:=FS[{x,y,-x*a-b*y}]
perpen[{a_,b_},{x_,y_,z_}]:=FS[{y,-x,x*b-y*a}]
parallelQ[{a_,b_,c_},{x_,y_,z_}]:=FS[a*y-x*b]
perpenQ[{a_,b_,c_},{x_,y_,z_}]:=FS[a*x+y*b]
```

When the functions `parallelQ` or `perpenQ`, for a given pair of lines, return the value zero, then these two lines are parallel or perpendicular, respectively.

In Mathematica the test for concurrency of three lines (i.e., whether they are parallel or intersect in a point) is the following.

```
concurQ[{a_,b_,c_},{i_,j_,k_},{p_,q_,r_}]:=
  FS[a*j*r-a*k*q-i*b*r+i*c*q+p*b*k-p*c*j]
```

Hence, three lines either intersect in a point or are parallel provided the value of the function `concurQ` in them is zero.

In solving problems using the analytic geometry it is often necessary to determine the projection of a point onto a line. Since the projection is the intersection

of the line and the perpendicular to the line through the point, if we input into Mathematica

```
P:={p, q}; m:={a, b, c}; Q:=inter[m,perpen[P,m]];
```

the output will be the coordinates of the projection  $Q$  of the point  $P$  onto the line  $m$ . Hence, the corresponding function looks as follows:

```
project[{p_, q_}, {a_, b_, c_}] :=
```

```
FS[{- (c*a+b*q*a-p*b^2)/(b^2+a^2), (-b*c+q*a^2-a*p*b)/(b^2+a^2)}]
```

This concludes the listing of the most basic functions for the analytic geometry of the plane. In the rest of this paper we shall present fifteen geometry problems from the problem collection [6] and give detailed solutions of them in Mathematica. The collection is for the first year high school level (age 15–16) but some solutions require knowledge from the second and the third year.

### 3. Fifteen problems

Our first example is the problem 395 from the book [6] that reads as follows:

**Problem 1.** Prove that the area  $P$  of a triangle  $ABC$  with vertices in the points  $A(x_1, y_1)$ ,  $B(x_2, y_2)$  and  $C(x_3, y_3)$  is given by the formula:

$$P = \frac{|x_1(y_2 - y_3) + x_2(y_3 - y_1) + x_3(y_1 - y_2)|}{2}$$

or

$$P = \frac{|y_1(x_2 - x_3) + y_2(x_3 - x_1) + y_3(x_1 - x_2)|}{2}.$$

**Solution.** Recall that the area of a triangle is a half of the product of lengths of any of its sides with the corresponding altitude. Hence, with the help of Mathematica functions introduced earlier, the area is easily computed as follows:

```
tA:={Subscript[x,1],Subscript[y,1]};
```

```
tB:={Subscript[x,2],Subscript[y,2]};
```

```
tC:={Subscript[x,3],Subscript[y,3]};
```

```
tD:=project[tC,line2[tA,tB]];
```

```
vP:=FS[distance[tA,tB]*distance[tC,tD]/2];
```

The output in Mathematica will be a rather complicated expression

$$\frac{1}{2} \sqrt{\frac{(-y_3x_1+x_3y_1+y_3x_2-x_2y_1-x_3y_2+x_1y_2)^2}{x_2^2-2x_2x_1+x_1^2+y_1^2-2y_1y_2+y_2^2}} \sqrt{x_2^2-2x_2x_1+x_1^2+y_1^2-2y_1y_2+y_2^2}.$$

As the computer is just a machine and we have not explained the nature of symbols representing the coordinates of the vertices, it will not cancel out the denominator in the first square root with the second square root even though they

are clearly identical. It also does not notice that the square root of the square in the numerator of the first square root is equal to the absolute value

$$|-y_3x_1 + x_3y_1 + y_3x_2 - x_2y_1 - x_3y_2 + x_1y_2|.$$

When we make these simplifications we shall obviously get the required formula.

It is interesting to note that without the absolute value the above formula computes the oriented area of the triangle  $ABC$ . If this triangle is positively oriented, i.e., if the movement  $ABCA$  is in the counterclockwise direction, then this real number will be positive and otherwise is negative. It will be zero if and only if the points  $A$ ,  $B$  and  $C$  are collinear.

The function that gives this oriented area in Mathematica is realized in the following input:

```
area[{a_, x_}, {b_, y_}, {c_, z_}] := FS[(x*c-b*x-a*z+a*y+b*z-c*y)/2]
```

The second example is the problem 425 from the same book [6].

**Problem 2.** Let  $ABC$  be a triangle and let  $U$ ,  $V$ ,  $W$  be midpoints of sides  $\overline{BC}$ ,  $\overline{CA}$  and  $\overline{AB}$ . The segments  $\overline{AU}$ ,  $\overline{BV}$  and  $\overline{CW}$  are called the **medians** of the triangle  $ABC$ . Prove analytically that the three medians intersect in a point that we call the **centroid** of the triangle and that the centroid divides each median in the ratio  $2 : 1$  counting from the vertex.

**Solution.** The proof on the computer, in Mathematica, begins by typing the following input:

```
tA := {Subscript[x, 1], Subscript[y, 1]};
tB := {Subscript[x, 2], Subscript[y, 2]};
tC := {Subscript[x, 3], Subscript[y, 3]};
tU := midpoint[tB, tC]; tV := midpoint[tC, tA]; tW := midpoint[tA, tB];
concurQ[line2[tA, tU], line2[tB, tV], line2[tC, tW]];
```

In amazingly short time the computer will output the value zero which proves that the medians intersect in a point. The coordinates of this point are revealed with the commands:

```
tG := inter[line2[tA, tU], line2[tB, tV]];
```

The point  $G$  has the coordinates  $(\frac{x_1+x_2+x_3}{3}, \frac{y_1+y_2+y_3}{3})$  so that we can immediately write down the Mathematica function which associates the centroid to a triangle:

```
centroid[{a_, x_}, {b_, y_}, {c_, z_}] := FS[(a+b+c)/3, (x+y+z)/3]
```

In order to prove the second claim of the problem we shall find the point that divides the median of the vertex  $A$  (i.e., the segment  $\overline{AU}$ ) in the ratio  $2 : 1$  counting from the vertex  $A$  and show that it coincides with the point  $G$  (the centroid of the triangle  $ABC$ ). The same argument could be repeated for the medians of the vertices  $B$  and  $C$ .

```
tT := ratiomn[tA, tU, 2, 1]; distance[tG, tT]
```

Since the returned value is zero, the points  $G$  and  $T$  coincide so that the proof of the problem is completed successfully.

The third example is the problem 989 also from the collection [6].

**Problem 3.** Prove that the midpoints of sides and the feet of the altitudes of a triangle lie on the same circle.

**Solution.** Without loss of generality we can assume that the points  $A$ ,  $B$  and  $C$  are selected in the plane so that their coordinates are  $(0, 0)$ ,  $(c, 0)$  and  $(u, v)$ , where  $c$ ,  $u$  and  $v$  are real numbers with  $c$  and  $v$  different from zero.

```
eA:={0, 0}; eB:={c, 0}; eC:={u, v};
```

Then we get the midpoints of the sides applying the function `midpoint`:

```
eAp:=midpoint[eB,eC]; eBp:=midpoint[eC,eA]; eCp:=midpoint[eA,eB];
```

The feet of the altitudes are the projections of the vertices onto the opposite sidelines:

```
eApp:=project[eA,line2[eB,eC]]; eBpp:=project[eB,line2[eC,eA]];
eCp:=project[eC,line2[eA,eB]];
```

The center of the circle circumscribed to a triangle is the intersection of perpendicular bisectors of its sides. Hence, in our situation, the center  $S$  of the circle circumscribed to the triangle  $A'B'C'$  with vertices in the midpoints of sides is defined as follows:

```
eS:=inter[perpen[midpoint[eBp,eCp],line2[eBp,eCp]],
           perpen[midpoint[eCp,eAp],line2[eCp,eAp]]]
```

Applying the same method to the triangle  $A''B''C''$  with vertices at the feet of the altitudes we can find the center  $T$  of its circumscribed circle.

```
eT:=inter[perpen[midpoint[eBpp,eCpp],line2[eBpp,eCpp]],
           perpen[midpoint[eCp,eApp],line2[eCp,eApp]]]
```

After we type in the above commands the computer will output the coordinates of the points  $S$  and  $T$ . We see that they are equal, so that the points  $S$  and  $T$  coincide.

In order to complete the proof it remains still to prove that the radii of the circumcircles of the triangles  $A'B'C'$  and  $A''B''C''$  are equal. This is checked in Mathematica with the following input:

```
FS[distance[eS,eCp]-distance[eT,eCp]]
```

Since the returned value is zero the proof is successfully accomplished.

With almost no effort we can now prove that the radius of the above circle (also known as the nine-point circle because it also goes through the midpoints of the segments joining vertices with the orthocenter) is equal to the half of the radius of the circle circumscribed to the triangle  $ABC$ . In order to check this using the same method as above we first find the coordinates of the center  $O$  of the circumcircle of  $ABC$

```
eO:=inter[perpen[midpoint[eB,eC],line2[eB,eC]],
```

```
perpen[midpoint[eC,eA],line2[eC,eA]]]
```

and request from Mathematica to compute the following:

```
FS[distance[e0,eC]/distance[eS,eCp]]
```

Of course, the result is the number two.

The fourth example are the problems 719 and 720 from the book [6].

**Problem 4.** Prove that if a triangle has two equal altitudes or two equal medians, then it is isosceles.

**Solution.** With the assumptions and the notation from the proof of the Problem 3, typing in

```
FS[distance[eA,eApp]^2-distance[eB,eBpp]^2]
```

we obtain  $\frac{c^3 v^2 (2u-c)}{(v^2+u^2-2uc+c^2)(v^2+u^2)}$ . Hence, if the altitudes  $AA''$  and  $BB''$  have the same lengths then  $u = \frac{c}{2}$  so that  $ABC$  is an isosceles triangle because the vertex  $C$  lies on the perpendicular bisector of the side  $AB$ .

Similarly we see that after typing into the program Mathematica

```
FS[distance[eA,eAp]^2-distance[eB,eBp]^2]
```

the output is  $\frac{3c(2u-c)}{4}$  that leads to the same conclusion for medians.

More complicated to prove is the Problem 721 from [6]. Our method of its proof assumes the knowledge of the trigonometric functions (the cotangent in particular).

**Problem 5.** Prove that a triangle is isosceles if and only if it has two equal angle bisectors.

**Solution.** In order to have simple expressions we shall assume that the vertices  $A$  and  $B$  and the incenter  $I$  (i.e., the center  $I$  of the circle inscribed to the triangle  $ABC$ ) have the coordinates  $(0, 0)$ ,  $(f+g, 0)$ , and  $(f, 1)$ , where  $f$  and  $g$  are positive real numbers. In fact, these are the cotangents of the halves of the angles  $A$  and  $B$ . In addition, we assumed that the inradius is equal to 1.

```
tA:={0, 0}; tB:={f+g, 0}; tI:={f, 1}; tJc:={f, 0};
```

If the points  $J_a, J_b, J_c$  are the projections of the incenter  $I$  onto the sides of  $ABC$ , then  $J_c$  has the coordinates  $(f, 0)$  while we get the coordinates of  $J_a$  as solutions of the following system of equations:

```
sys:=Solve[{distance[tB,{p, q}]==distance[tB,tJc],
            distance[tI,{p, q}]==1},{p, q}];
```

where  $p$  and  $q$  are the coordinates of the point  $J_a$  that we are trying to determine. This system has only two solutions. The first are the coordinates of the point  $J_c$  while the second are the required coordinates  $\frac{f(g^2+1)+2g}{g^2+1}$  and  $\frac{2g^2}{g^2+1}$  of the point  $J_a$ .

```
tJa:={p,q} /. Extract[sys, 2]
```

In a similar way we can find also the coordinates  $\frac{f(f^2-1)}{f^2+1}$  i  $\frac{2f^2}{f^2+1}$  of the point  $J_b$ .

```
tJb:={p,q} /. Extract[Solve[{distance[tA,{p,q]}==
distance[tA,tJc], distance[tI,{p,q]}==1},{p,q}], 2]
```

Now we can find the points  $A_i$  and  $B_i$  of intersection of bisectors of angles  $A$  and  $B$  with the opposite sides as intersections  $AI \cap BJ_a$  and  $BI \cap AJ_b$ .

```
tAi:=inter[line2[tA,tI],line2[tB,tJa]];
```

```
tBi:=inter[line2[tB,tI],line2[tA,tJb]];
```

Let us now ask the program Mathematica to calculate the difference of the squares of lengths of angle bisectors with the following input:

```
Q:=FS[distance[tA,tAi]^2-distance[tB,tBi]^2];
```

The output will be the quotient

$$\frac{4(f+g)^3(f-g)(f^4g^2+4g^3f^3-5f^2g^2+g^4f^2+4fg-1)}{(g^2+2fg-1)^2(f^2+2fg-1)^2}.$$

Since its numerator contains  $f-g$  as a factor and  $f+g$  is obviously never zero, we conclude that the proof will be completed provided we show that the long parenthesis

$$Z = f^4g^2 + 4g^3f^3 - 5f^2g^2 + g^4f^2 + 4fg - 1$$

in the numerator is always positive.

First note that the sum  $\frac{A}{2} + \frac{B}{2}$  of halves of the angles is at most  $\frac{\pi}{2}$  so that

$$\cot\left(\frac{A}{2} + \frac{B}{2}\right) = \frac{\cot(\frac{A}{2})\cot(\frac{B}{2}) - 1}{\cot(\frac{A}{2}) + \cot(\frac{B}{2})} = \frac{fg - 1}{f + g} > 0.$$

We conclude that  $fg > 1$ .

The first and the fourth term of  $Z$  together give

$$f^4g^2 + f^2g^4 = (f^2 + g^2)(fg)^2 \geq 2(fg)(fg)^2 = 2(fg)^3$$

because  $f^2 + g^2 \geq 2fg$ . If we introduce the notation  $\vartheta = fg$  then

$$Z \geq 6\vartheta^3 - 5\vartheta^2 + 4\vartheta - 1.$$

Since  $\vartheta > 1$  we can replace  $\vartheta$  in the above cubic polynomial with  $1 + \eta$  with  $\eta > 0$  and get  $(3\eta + 2)(2\eta^2 + 3\eta + 2)$ . This expression is always positive because the new variable  $\eta$  is positive. This completes the proof.

Notice that the same could be obtained with the substitution  $f = \frac{1+k}{g}$  for the positive real number  $k$  in the polynomial  $Z$ . Following the input

```
Collect[Extract[Q,4] /. f->(1+k)/g, g];
```



the program Mathematica outputs

$$(1+k)^2 g^2 + \frac{(1+k)^4}{g^2} + 2 + 6k + 7k^2 + 4k^3$$

which is obviously always positive.

We continue with the problem 833 from [6] which is in the section about similarity of triangles.

**Problem 6.** Let  $r$  be the radius of the circle inscribed to a triangle  $ABC$  and let  $R$  be the radius of its circumscribed circle. Prove that  $R \geq 2r$ .

**Solution.** The following proof has great similarity with the solution of the previous problem. Without loss of generality we can assume that the angles  $A$  and  $B$  of the triangle  $ABC$  are acute (i.e. less than  $\frac{\pi}{2}$  radians) and that the vertices  $A$ ,  $B$  and the center  $I$  of the incircle have the coordinates  $(0, 0)$ ,  $(r(f+g), 0)$  and  $(fr, r)$  for some real numbers  $f > 1$ ,  $g > 1$  and  $r > 0$ .

Our idea of the proof is first to determine the coordinates of the vertex  $C$  and the center  $O$  of the circumcircle. This will make it possible to compute the radius  $R$  of the circumcircle. Finally, we show that the difference  $R - 2r$  is always positive except in the case of the equilateral triangle when it is zero.

Let  $J_a$ ,  $J_b$ ,  $J_c$  be projections of the center  $I$  of the incircle onto the sides of the triangle  $ABC$ . The point  $J_c$  has the coordinates  $(fr, 0)$  while we get the coordinates of the  $J_a$  from the following system of the equations

```
sys:=Solve[{distance[tB,{p,q]}==
            distance[tB,tJc], distance[tI,{p,q]}==r},{p,q}];
```

where  $p$  and  $q$  are the wanted coordinates of the point  $J_a$ . This system has two solutions: the coordinates of the point  $J_c$  and the coordinates  $\frac{(g^2 f + 2g + f)r}{1+g^2}$  and  $\frac{2rg^2}{1+g^2}$  of  $J_a$ . In a similar way we get the coordinates  $\frac{f(f^2-1)r}{f^2+1}$  and  $\frac{2f^2 r}{f^2+1}$  of the point  $J_b$ .

```
tJa:={p,q} /. Extract[sys, 2]
tJb:={p,q} /. Extract[Solve[{distance[tA,{p,q]}==
            distance[tA,tJc], distance[tI,{p,q]}==r},{p,q}], 2]
```

The vertex  $C$  is the intersection  $AJ_b \cap BJ_a$ .

```
tC:=inter[line2[tA,tJb],line2[tB,tJa]];
```

The center  $O$  of the circumcircle and its radius  $R$  are given as the solutions of the following system of equations:

```
tO:={p,q}; Solve[{distance[tA,tO]==R, distance[tB,tO]==R,
            distance[tC,tO]==R},{p,q,R]};
```

From the two solutions of the system only the one where

$$R = \frac{r(1+g^2)(1+f^2)}{4(fg-1)}$$

is correct. In the second solution the radius  $R$  is negative which is not acceptable.

```
R:=r*(1+f^2)*(1+g^2)/4/(f*g-1);
```

```
M:=Collect[Extract[FS[R-2*r],2],f]; \[CapitalDelta]:=
```

```
FS[Coefficient[M,f,1]^2-4*Coefficient[M,f,2]*Coefficient[M,f,0]]
```

The difference  $R - 2r$  is equal to  $\frac{Mr}{4(fg-1)}$ , where  $M$  is the quadratic trinomial

$$(g^2 + 1)f^2 - 8gf + g^2 + 9$$

in  $f$ . Its discriminant is  $-4(-3+g^2)^2$  which is always negative (so that  $M > 0$  because the leading coefficient  $g^2 + 1$  is positive) except when  $g = \cot \frac{B}{2} = \sqrt{3}$  and  $f = \sqrt{3}$  (i.e. the triangle  $ABC$  is equilateral) when  $M = 0$ .

Next is the problem 312 from [6] which is in the second chapter on the perimeter and the area of circles.

**Problem 7.** Let  $T$  be a point inside the triangle  $ABC$  and let  $A_1, B_1, C_1$  be interior points of the sides  $\overline{BC}, \overline{CA}, \overline{AB}$ . Let  $R_i$  for  $i = 1, 2, 3, 4, 5, 6$  be radii of the circumcircles of the triangles  $AC_1T, C_1BT, BA_1T, A_1CT, CB_1T, B_1AT$ . Prove that  $R_1 R_3 R_5 = R_2 R_4 R_6$ .

**Solution.** Let us first define in Mathematica the function which associates to a given triple of points the radius of the circumcircle of the triangle whose vertices are these points.

```
bisector[a_, b_] :=perpen[midpoint[a,b],line2[a,b]];
```

```
CC[a_, b_, c_] :=inter[bisector[a,b],bisector[a,c]];
```

```
RC[a_, b_, c_] :=distance[a,CC[a,b,c]];
```

Let us now input the points  $A, B, C$  and  $T$ .

```
tA:={0,0}; tB:={c,0}; tC:={s,t}; tT:={p,q};
```

If  $s \neq c$  then the position of a point  $A_1$  on the line  $BC$  can be described by a real number  $u$  and the coordinates of this point are  $\left(u, \frac{t(c-u)}{c-s}\right)$ . We get this by requiring that the point with the coordinates  $(u, z)$  lies on the line  $BC$  and then solve the condition with respect to  $z$ .

```
tA1:={u,z} /. Solve[onlineQ[{u,z},line2[tB,tC]]==0,z];
```

Similarly, if  $s \neq 0$ , then any point  $B_1$  on the line  $CA$  has the coordinates  $\left(v, \frac{tv}{s}\right)$  and any point  $C_1$  on the line  $AB$  has the coordinates  $(w, 0)$  for some real numbers  $v$  and  $w$ .

```
tA1:={u,t*(c-u)/(c-s)}; tB1:={v,t*v/s}; tC1:={w,0};
```

If  $s = c$  then any point  $A_2$  on the line  $BC$  has the coordinates  $(c, u)$  for some real number  $u$ . If  $s = 0$  then any point  $B_2$  on the line  $CA$  has the coordinates  $(0, v)$  for some real number  $v$ .

`tA2:={c,u}; tB2:={0,v};`

Let us now define a function which computes the difference of the squares of the products of radii of the circumcircles for seven points in the plane.

`FR[a_, b_, c_, d_, e_, f_, g_] := FS[(RC[a, f, g]*RC[b, d, g]*RC[c, e, g])^2 - (RC[f, b, g]*RC[d, c, g]*RC[e, a, g])^2];`

It is now easy to check that the following values are zero:

`FR[tA, tB, tC, tA1, tB1, tC1, tT]`

`s:=c; FR[tA, tB, tC, tA2, tB1, tC1, tT]`

`s:=0, FR[tA, tB, tC, tA1, tB2, tC1, tT]`

This completes the solution of the Problem 312 from [6] in the program Mathematica.

**Remark.** It is clear from the above proof that we have never used the assumption that the point  $T$  is inside of the triangle  $ABC$  nor the assumption that the points  $A_1, B_1, C_1$  are interior points of the sides  $\overline{BC}, \overline{CA}, \overline{AB}$ . In this way, using the computer, we succeeded to prove a more general statement.

The following example is the Problem 644 from the collection [6] which is in the section on the volume of the cylinder, cone, and ball.

**Problem 8.** On the bottom of the cylindrical container whose base has the diameter 15 cm there is a ball with the diameter 12 cm. The water is poured into the container up to the highest point of the ball. For how many cm will drop the level of the water when the ball is taken out?

**Solution.** Recall the formulas  $V_B = \frac{4}{3} \left(\frac{D}{2}\right)^3 \pi$  for the volume of the ball with the diameter  $D$  and  $V_C = \left(\frac{d}{2}\right)^2 h \pi$  for the volume of the cylinder of the height  $h$  whose base is a circle with the diameter  $d$ .

In the program Mathematica these volume functions are defined as follows:

`VB[d_]:=d^3*Pi/6; VC[d_, h_]:=d^2*h*Pi/4;`

The volume of the water in the container is the difference of the volume of the cylinder (with the height equal to the diameter of the ball) and the volume of the ball:

`Vwater:=VC[15, 12]-VB[12];`

After the removal of the ball the water will fill in the cylindrical container whose base is the circle with the diameter of 15 cm and its height will be  $12 - p$  cm where  $p$  is the required drop in the level of the water in the container. This drop  $p$  is found in the program Mathematica as follows:

`Solve[Vwater==VC[15, 12-p], p]`

The solution is  $p = 5.12$  cm.

Another nice example is the Problem 963 from [6]. We assume again the knowledge of trigonometric functions.

**Problem 9.** A trapezium is circumscribed about the circle with the radius  $R$ . The chord that joins the touching points of the lateral sides has the length  $b$  and is parallel to the bases. Prove that the area of the trapezium is  $\frac{8R^3}{b}$ .

**Solution.** Select the rectangular coordinate system so that the circle  $k$  with the radius  $R$  inscribed to the trapezium  $ACEG$  has the center in the origin and its parallel sides (bases)  $AC$  and  $EG$  touch  $k$  in the points  $B(0, -R)$  and  $F(0, R)$ . Let the lateral sides  $CE$  and  $AG$  touch  $k$  in the points  $D(R \cos \theta, R \sin \theta)$  and  $H(R \cos \sigma, R \sin \sigma)$  for some angles  $\theta$  and  $\sigma$ .

Let us first input into the program Mathematica the points  $O, B, F, D, H$  and the lines  $AC, EG$ .

```
tO:={0, 0}; tB:={0, -R}; tF:={0, R};
tD:={R Cos[\[Theta]], R Sin[\[Theta]]};
tH:={R Cos[\[Sigma]], R Sin[\[Sigma]]};
pAC:={0, 1, R}; pEG:={0, 1, -R};
```

Then we ask when will the chord  $DH$  joining the touching points  $D$  and  $H$  of the lateral sides be parallel with the bases.

```
parallelQ[line2[tD, tH], pAC]
```

The condition is  $R(\sin \theta - \sin \sigma) = 0$  so that we must have  $\sigma = \pi - \theta$ . Hence, the trapezium  $ACEG$  is equilateral and symmetrical with respect to the line  $BF$ . It suffices therefore to find the area only of the right half  $BCEF$ .

The line  $CE$  is the perpendicular in the point  $D$  to the line  $OD$  (the property of the tangent to the circle) and the points  $C$  and  $E$  are the intersections of the line  $CE$  with the lines  $AC$  and  $EG$ .

```
pCE:=perpen[tD, line2[tO, tD]];
tC:=inter[pAC, pCE]; tE:=inter[pEG, pCE];
```

The area of the right half  $BCEF$  is the sum of the areas of the triangles  $BCE$  and  $BEF$ .

```
FS[area[tB, tC, tE]+area[tB, tE, tF]]
```

The program Mathematica will compute that this sum has the value  $\frac{2R^2}{\cos \theta}$ . Since  $b = 2R \cos \theta$ , we conclude that the wanted area of the trapezium  $ACEG$  is indeed  $\frac{8R^3}{b}$ .

**Remark.** In the book [6] there is the incorrect claim that the area of the trapezium is  $\frac{4R^3}{b}$ . Using the approach from the solution of the Problem 13 (i.e., the Problem 1112 from [6]) it is possible to completely avoid the trigonometric functions. This solution we leave to the readers as an exercise.

We continue with the solution of the Problem 1026 from [6].

**Problem 10.** Prove that in every regular heptagon  $A_1A_2A_3A_4A_5A_6A_7$  the following equality holds:

$$\frac{1}{|A_1A_2|} = \frac{1}{|A_1A_3|} + \frac{1}{|A_1A_4|}.$$

**Solution.** Choose the coordinate system so that the circle  $k$  with the center at the origin and with the radius  $R$  is circumscribed to the heptagon  $A_1A_2A_3A_4A_5A_6A_7$ . We can assume that the vertex  $A_1$  has the coordinates  $(R, 0)$ . The other relevant vertices have the coordinates  $A_2 (R \cos \frac{2\pi}{7}, R \sin \frac{2\pi}{7})$ ,  $A_3 (R \cos \frac{4\pi}{7}, R \sin \frac{4\pi}{7})$ ,  $A_4 (R \cos \frac{6\pi}{7}, R \sin \frac{6\pi}{7})$ .

Let us input these points into the program Mathematica:

```
tA1:={R, 0}; tA2:={RCos[2 Pi/7], RSin[2 Pi/7]};
tA3:={RCos[4 Pi/7], RSin[4 Pi/7]};
tA4:={RCos[6 Pi/7], RSin[6 Pi/7]};
```

In order to check the above relation among the reciprocal values we must type into the program Mathematica the following:

```
FullSimplify[Numerator[Together[1/distance[tA1, tA2]-
1/distance[tA1, tA3]-1/distance[tA1, tA4]]], R>0]
```

For few seconds the computer will output the value zero which proves that the statement in the problem holds.

**Remark.** Several other interesting properties of the regular heptagon proved in the program Maple V are described in the article [5].

Next is the Problem 1084 from the section eight of the collection [6].

**Problem 11.** The projections of the legs of the right triangle onto the hypotenuse have lengths  $\frac{18}{5}$ ,  $\frac{32}{5}$ . Find the radius of the circle inscribed into this triangle?

**Solution.** Select the rectangular coordinate system so that its origin is the vertex  $C$  of the right triangle and its legs are on the coordinate axes. We can assume that the remaining vertices  $A$  and  $B$  have the coordinates  $(0, b)$  and  $(a, 0)$ , for some positive real numbers  $a$  and  $b$ .

In the program Mathematica these points are input as follows:

```
tC:={0, 0}; tA:={0, b}; tB:={a, 0};
```

Then we find the projection  $D$  of the vertex  $C$  onto the hypotenuse  $AB$ .

```
tD:=project[tC, line2[tA, tB]];
```

The values for the variables  $a$  and  $b$  can be determined from the information that  $|AD| = \frac{18}{5}$  and  $|BD| = \frac{32}{5}$ .

```
Solve[{distance[tA, tD]==18/5, distance[tB, tD]==32/5}, {a, b}];
```

There are eight solutions (four real and four complex) but only one when  $a = 8$  and  $b = 6$  is acceptable. Hence, this right triangle has sides 8, 6, 10 (that are twice

as long as the sides of the standard (Egyptian) right triangle with sides 4, 3, 5) so that its inscribed circle has the radius  $r = 2$ .

This could also be seen by asking that the center  $I$  of the inscribed circle with the coordinates  $(r, r)$  is at the distance  $r$  from the line  $AB$ .

```
a:=8; b:=6; tI:={r, r};
```

```
Solve[distance[tI, project[tI, line2[tA,tB]]]==r, r];
```

>From the two solutions  $r = 2$  and  $r = 12$  only the first satisfies the conditions of the problem. The second solution gives the radius of the corresponding excircle.

Now we consider the Problem 1103 again from the collection [6].

**Problem 12.** Two sides of the triangle have the length 6 cm and 8 cm. The medians of these sides are perpendicular. Find the third side of this triangle.

**Solution.** Let the triangle  $ABC$  be embedded into the rectangular coordinate system so that  $A(0, 0)$ ,  $B(c, 0)$ , and  $C(u, v)$  for positive real numbers  $c$  and  $v$  and for a real number  $u$ .

In the program Mathematica these points and the centroid  $T$  are input as follows:

```
tA:={0, 0}; tB:={c, 0}; tC:={u, v}; tT:=centroid[tA,tB,tC];
```

Since the medians of the vertices  $A$  and  $B$  are perpendicular,  $ABT$  is the right triangle and  $c^2 = |AB|^2 = |AT|^2 + |BT|^2$  by the Pythagorean theorem. On the other hand  $|BC| = 6$  and  $|AC| = 8$ . If we ask the program Mathematica to solve this system of three equations in the variables  $c$ ,  $u$ , and  $v$  with the input

```
Solve[{distance[tB,tC]==6, distance[tA,tC]==8,
      c^2==distance[tA,tT]^2+distance[tB,tT]^2}, {c, u, v}]
```

it will respond with two solutions. Only the one where  $c = 2\sqrt{5}$  cm is correct.

Our next example is the Problem 1112 from [6].

**Problem 13.** A circle is inscribed into a trapezium. Prove that the ratio of the areas of the circle and the trapezium is equal to the ratio of their perimeters.

**Solution.** Choose the rectangular coordinate system so that the circle  $k$  with the radius  $R$  which is inscribed to the trapezium  $ACEG$  has the center in the origin while its parallel sides (bases)  $AC$  and  $EG$  touch  $k$  in the points  $B(0, -R)$  and  $F(0, R)$ . Let the vertices  $A$  and  $C$  have the coordinates  $(-u, -R)$  and  $(v, -R)$  for positive real numbers  $u$  and  $v$ . Let the lateral sides  $CE$  and  $AG$  touch  $k$  in the points  $D$  and  $H$ . Our first goal is to find the coordinates of these points and then the coordinates of the vertices  $E$  and  $G$ .

Let us first input into the program Mathematica the points  $O, B, F, A, C$  and the lines  $AC, EG$ .

```
tO:={0, 0}; tB:={0, -R}; tF:={0, R}; tA:={-u, -R};
```

```
tC:={v, -R}; pAC:={0, 1, R}; pEG:={0, 1, -R};
```

Assume that the point  $H$  has the coordinates  $(p, q)$ . They must satisfy two conditions. The first is  $p^2 + q^2 = R^2$  i.e. that the point  $H$  lies on the circle  $k$ . The second condition is that the distance from  $A$  to  $H$  is equal to  $u$  because the lines  $AB$  and  $AH$  are tangents through the point  $A$  onto the circle  $k$ .

```
H:=Solve[{p^2+q^2==R^2, distance[{p, q}, tA]==u}, {p, q}]
tH:={p, q} /. H
```

In a similar way we can determine the coordinates of the point  $D$ .

```
K:=Solve[{p^2+q^2==R^2, distance[{p, q}, tC]==v}, {p, q}]
tD:={p, q} /. K
```

The vertices  $E$  and  $G$  are the intersections of the line  $EG$  with the lines  $CD$  and  $AH$ , respectively.

```
pAH:=line2[tA, tH]; pCD:=line2[tC, tD];
tE:=inter[pEG, pCD]; tG:=inter[pEG, pAH];
```

The first coordinates of the points  $E$  and  $G$  are  $\frac{R^2}{v}$  and  $-\frac{R^2}{u}$ . Hence, the perimeter  $O_{ACEG}$  of the trapezium  $ACEG$  is  $2(u + v + \frac{R^2}{u} + \frac{R^2}{v})$ . Its area  $P_{ACEG}$  is

```
FS[area[tA, tC, tE]+area[tA, tE, tG]]
```

equal to  $\frac{R(u+v)(u v + R^2)}{u v}$ . Now it is easy to check that

$$\frac{2 R \pi}{O_{ACEG}} = \frac{R^2 \pi}{P_{ACEG}}.$$

**Remark.** In [6] there are no solutions for the Problem 1112.

The next example is the Problem 1139 from [6].

**Problem 14.** Prove that if the angle bisector of a triangle is also the bisector of the angle determined by the altitude and the median, then this triangle is right.

**Solution.** Let us choose the rectangular coordinate system so that the points  $A(0, 0)$ ,  $B((f + g)r, 0)$ ,  $C\left(\frac{rg(f^2-1)}{fg-1}, \frac{2fgr}{fg-1}\right)$  are the vertices of the triangle and the center of its inscribed circle is the point  $I(fr, r)$ , where  $f$  and  $g$  are cotangents of  $\frac{A}{2}$  and  $\frac{B}{2}$  and  $r$  is the radius of the incircle.

We shall first input into the program Mathematica the points  $A, B$ , the midpoint  $C_g$  of the segment  $AB$ , the points  $C, I$  and the feet  $C_h$  of the altitude of the vertex  $C$  on the line  $AB$ .

```
tA:={0, 0}; tB:={r*(f+g), 0}; tCg:=midpoint[tA,tB];
tC:={r*g*(f^2-1)/(f*g-1), 2*f*g*r/(f*g-1)}; tI:={f*r, r};
tCh:=project[tC,line2[tA,tB]];
```

In order that the bisector of the angle  $C$  (i.e. the line  $CI$ ) is the bisector of the angle between the altitude (i.e. the line  $CC_h$ ) and the median (i.e. the line  $CC_g$ )

it is necessary and sufficient that the segments  $II_h$  and  $II_g$  have the same length, where  $I_h$  and  $I_g$  are the projections of the point  $I$  onto the lines  $CC_h$  and  $CC_g$ .

```
tIh:=project[tI,line2[tC,tCh]]; tIg:=project[tI,line2[tC,tCg]];
IZ:=FS[distance[tI,tIg]^2-distance[tI,tIh]^2]
```

The program Mathematica reports that the expression  $IZ$  is equal

$$\frac{r^2 (f-g)^2 (fg+g+f-1)(fg-g-f-1)(fg+1)^2}{(12f^2g^2+g^2f^4-2f^3g^3+g^4f^2+2f^3g+2fg^3+f^2-2fg+g^2)(fg-1)^2}.$$

Hence, it will be zero if and only if  $f = g$  (i.e.  $|BC| = |CA|$  so that the triangle  $ABC$  is isosceles) or

$$(fg+g+f-1)(fg-g-f-1) = 0$$

which is the condition for the lines  $BC$  and  $CA$  to be perpendicular (i.e. that the angle  $C$  has 90 degrees and the triangle  $ABC$  is right).

```
perpenQ[line2[tB,tC],line2[tC,tA]]
```

**Remark.** In the collection [6] the possibility that the triangle  $ABC$  is isosceles is absent.

Our final example is the Problem 1152 from [6].

**Problem 15.** Let different points  $A$  and  $B$  be given and let the point  $T$  be outside the line  $AB$ . Through the point  $T$  construct the line  $m$  so that the ratio of the distances of the points  $A$  and  $B$  to the line  $m$  is  $2 : 3$ .

**Solution.** Choose the rectangular coordinate system so that the given points are  $A(0, 0)$ ,  $B(c, 0)$ , and  $T(p, q)$  for real numbers  $c, p, q$ . Let the line  $m$  has the equation  $ux + vy + w = 0$  for some real numbers  $u, v, w$ . In order that it goes through the point  $T$  the free term  $w$  must be equal to  $-up - vq$ .

Let us input into the program Mathematica the points  $A, B, T$  and the line  $m$ .

```
tA:={0, 0}; tB:={c, 0}; tT:={p, q}; pm:={u, v, -u*p-v*q};
```

Let  $A_m$  and  $B_m$  be the projections of the points  $A$  and  $B$  onto the line  $m$ .

```
tAm:=project[tA, pm]; tBm:=project[tB, pm];
```

By the requirement of the problem the quotient  $\frac{|AA_m|}{|BB_m|}$  is equal to  $\frac{2}{3}$ . Notice that the expression

```
IZ:=FS[distance[tA,tAm]^2/distance[tB,tBm]^2-4/9]
```

has as the numerator the product  $(5up + 5vq - 2uc)(up + vq + 2uc)$ . Hence, there are two possibilities  $q = \frac{-u(2c+p)}{q}$  and  $q = \frac{u(2c-5p)}{q}$ . They give lines  $qx - (2c+p)y + 2qc = 0$  and  $5qx + (2c-5p)y - 2qc = 0$  as solutions of the problem. Even though we know the solutions the question remains how to construct them. But, it is simple



to see that they intersect the line  $AB$  in the points  $C(-2c, 0)$  and  $D(\frac{2}{3}c, 0)$  and these are easily constructed.

**Remark.** In the collection [6] there are no solutions for the problem 1152.

**Remark.** A longer version of this paper with figures is available on the Internet at the home page of the second author: <http://www.math.hr/~cerin>

## References

- [1] BATOR, M., ČERIN, Z., ČULAV, M., Analitička geometrija ravnine računalom, *Matematičko-fizički list* **54** (2003/2004), 26–36.
- [2] BATOR, M., ČERIN, Z., ČULAV, M., Analitička geometrija ravnine i Matematika, *Prim. Math.* (to appear).
- [3] ČERIN, Z., Šest zadataka riješenih programom Maple V, *Matematičko-fizički list* **54** (2003/2004), 105–112.
- [4] ČERIN, Z., Još šest zadataka riješenih programom Maple V, *Matematičko-fizički list* **54** (2003/2004), 194–200.
- [5] ČERIN, Z., Geometrija pravilnog sedmerokuta, *Poučak* **14** (2003), 5–14.
- [6] PAVKOVIĆ, B., VELJAN, D., *Matematika 1*, Školska knjiga, Zagreb, 1999.

## Maja Bator

Odranska 8  
10000 Zagreb  
Croatia  
E-mail: batormaja@yahoo.com

## Zvonko Čerin

Kopernikova 7  
10000 Zagreb  
Croatia  
E-mail: cerin@math.hr

## Milena Čulav

Lhotkina 3  
10000 Zagreb  
Croatia  
E-mail: mickey@student.math.hr